

---

**清達光電技術有限公司**  
**TSINGTEK DISPLAY CO.,LTD**

**LCD CONTROLLER & DRIVER**  
**LC7981 OR EQUIVALENT**

---

公司总部： 杭州市费家塘路588号下城区高新技术产业基地4幢1楼  
0571-85121224 85121742 85121304(f) (总公司)  
杭州门市： 杭州市登云路639号杭州电子市场2A095  
0571-88256346 89902095 (f)  
北京分公司： 北京市海淀区中关村新中发电子市场B1618  
010-62051209 62000662 62568913 82036512 (f)  
网址： <http://www.tsingtek.com>  
E-mail: [sales@tsingtek.com](mailto:sales@tsingtek.com)



## 目 录

第一章 液晶显示模块汇总	3
第二章 液晶模块的供电说明	4
第三章 LC7981的一般介绍	5
第四章 内藏LC7981型液晶模块的应用	10
第五章 液晶模块的调试经验	20
附录一 清达光电配套产品	20
附录二 清达光电维修服务规范	20
附录三 LCM的常见故障	21



## 第一章 液晶显示模块汇总

### 1. 清达光电液晶命名规则

HG 16080 1 - - - - -

	产品序列号：HC 点阵字符；HG→SMT/COB图形单色；HGT→TAB图形单色；HGO→COG图形单色；HGR→COLOR STN；HGO TFT；HGS OLED
	字符 → 字符数（每行字符数*行数） 图形 → 点阵数
	产品序列
	显示模式：省略→STN黄绿模式；G→STN灰模式；B→STN蓝模式； F→FSTN半透半反；T→FSTN透射
	背光类型：省略 → 无背光； LY LED黄绿底光；SY LED黄绿侧光； LW LED白光；SW LED高亮白光；LB LED蓝光； LR LED红光；LA LED琥珀光；LG LED绿光； EB EL蓝光；EG EL绿光；EW EL白光； CW CCFL白光
	温度范围：省略 常温；H→宽温；EH→特宽温
	电源：省略 5V单电源；NV 5V双电源；SV 5V带温度补偿； LV 3/3.3V单电源；LNV 3/3.3V双电源；LSV 3/3.3V带温度补偿； OV 5V单电源，V0脚空脚（部分型号PCB板上有电位器可微调对比度） LOV 5V单电源，V0脚空脚（部分型号PCB板上有电位器可微调对比度）
	背光输入电压说明，请参照液晶详细资料
	特殊编号：TP 带触摸屏；S 串行通信；D 分屏
	产品内部编号

备注：\*具体背光电压值请参照液晶详细资料

### 2. 清达光电液晶汇总

清达光电出品的内置LC7981控制器及兼容型控制器图形液晶显示模块现按照完整型号汇总成表，便于选型和查询。



型号	模块尺寸 W × H(mm)	显示尺寸 W × H(mm)	背光			显示模式			
			LED	EL	CCFL	STN GRAY	STN YELLOW	STN BLUE	FSTN
HG128128B	72.5 × 69.9	50.0 × 49.0							
HG16080A	93.0 × 70.0	72.0 × 40.0							
HG160801	100.0 × 55.0	72.0 × 40.0							
HG160160B	85.0 × 100.0	62.0 × 62.0							
HG24064B	180.0 × 65.0	133.0 × 39.0							
HG240128D	144.0 × 104.0	114.0 × 64.0							

## 第二章 液晶模块的供电说明

- 模块应用一般有三种电源：逻辑电压、液晶驱动电压、背光电压。
- 如果您选用的液晶模块是单电源供电，分三种情况
  - 单电源带温度补偿-SV型(液晶型号后缀带-SV标志),那么V0脚要悬空
  - 单电源-OV型(液晶型号后缀带-OV标志),那么V0脚要悬空
  - 普通单电源型 负压由Vee输出，通过10K电位器输入至V0来调节对比度，具体接法如下：




具体电源接法请参照每款液晶引脚说明

- 如果您所选用的液晶模块是双电源（VDD/V0）供电，需要提供一个液晶驱动电压（即Vadj/VEE, 一般为负），用以调节对比度，接在液晶模块的Vadj/VEE引脚上。因为液晶材料的物理特性，液晶的对比度会随着温度的变化而相应变化，所以，您加的液晶驱动电压值应该随温度作相应的调整，大致是温度变化10℃，电压变化0.5伏左右。为满足这一要求，您要选择较大值的电源，然后做一个温度补偿电路，或者安排一个电位器，调整液晶驱动电压值。如果您提供的电源超过了液晶驱动电源的极限值，应该考虑保护电路。
- 常用驱动电源产生办法：
  - 清达光电提供的两种DC-DC变换器，是通用LCD驱动可调电源，可产生连续可



调负电压-1.5V - -30V电压或正电压0V - +30V电压；

- 2) 采用DC-DC模块，市场上常见的5D\*\*系列型号可选择使用；
- 3) 采用DC-DC集成电路制作负电源，如MAX749、MAX680、MC34063等；
- 4) 采用79系列三端集成稳压器，可产生-18v(7918)、-24v(7924)等电源。
5. 在您调试液晶模块时，请注意正确接线，尤其是逻辑电压、液晶驱动电源的接线不能有错，否则烧电路上的芯片。为避免液晶模块的损坏，在加液晶驱动电压（V0/VEE）时，需要比加逻辑电压（VDD）滞后50 ms；在关电时，液晶驱动电压（V0/VEE）需要比逻辑电压（VDD）提前50 ms关断。
6. 对于LED黄绿背光方式，供电为3.8-4.2V直流电源，**严格限制5V电源直接供电**（除非有特别说明），否则不仅会增加您的功耗，更会增加损坏背光灯的可能性和缩短液晶模块的使用寿命。
7. 如果采用CCFL背光方式，驱动电压将是交流1000V左右，需要用到逆变器，清达光电有配套逆变器提供，接法如图二（**OUT1和OUT2不能连在一起使用**）。注意逆变器不能空载使用，否则会自烧。



图二 CCFL逆变器接线图

### 第三章 LC7981的一般介绍

LC7981是一种图形液晶显示控制器，为HD61830的兼容控制器，它可与8位微处理器接口，管理64K显示RAM，内部时序发生器产生点阵液晶显示驱动信号。本手册将对LC7981作一般介绍，然后详细叙述内藏LC7981控制器的液晶显示模块的应用。

#### 一、LC7981的特点

- 1、LC7981是图形液晶显示控制器，可直接与M6800系列时序的MPU接口；
- 2、具有专用指令集，可完成文本显示或图形显示的功能设置，以及实现画面卷动、光标、闪烁、位操作等功能；
- 3、LC7981可管理64K显示RAM，其中图形方式为64K，字符方式为4K；
- 4、内部字符发生器CGROM共有192种字符，其中5×7字体160种，5×11字体有32种，HD61830还可外接字符发生器，使字符量达到256种；
- 5、LC7981可以静态方式显示至1/128占空比的动态方式显示。




## 二、LC7981的引脚说明及其功能

LC7981封装为60个引脚，管脚图和功能原理图分别如附录一和附录二所示。

管脚功能如下：

符号	状态	名称	功 能
DB0 - DB7	三态	与MPU接口的数据总线	
/CS	输入	片选信号	低电平有效
R/W	输入	读、写选择信号	当R/W=1时MPU从LC7981读出数据，当R/W=0时MPU向LC7981写数据
RS	输入	寄存器选择	RS=1选通指令寄存器，RS=0选通数据寄存器
E	输入	使能信号	在E下降沿写数据，E为高电平时为读数据
CR,R,C		RC振荡器引出端	接入振荡电阻和电容即可起振，使LC7981工作
RES	输入	复位信号	低电平有效
MA0 - MA15	输出	外接显示RAM的地址输出端	文本方式下，MA12 - MA15为外接字符库CGROM的地址输出端，所以LC7981只能管理4K的文本显示RAM
MD0 - MD7	三态	显示RAM数据总线	
RD0 - RD7	输入	外接CGROM数据输入端	
WE	输出	显示RAM的写信号	
CL1,CL2	输出	LCD驱动器的显示数据的锁存信号和移动时钟	
FLM	输出	显示同步器帧信号	
MA,MB	输出	液晶交流驱动信号	
D1,D2	输出	显示数据串行输出	液晶屏为单屏结构时，只使用D1；当液晶屏为双屏结构时，D1、D2分别用于上、下半屏
CP0	输出	时钟信号输出端	
SYNC	三态	多片HD61830并联使用时的同步信号	

## 三、与MPU接口的读、写操作时序，如下图所示：



当/CS为低电平时，RS、R/W、E的各种组合所实现的功能如下表：



RS	R/W	E	功能
0	0	下降沿	写数据或指令参数
0	1	高电平	读数据
1	0	下降沿	写指令代码
1	1	高电平	读忙标志位

## 四、指令系统

LC7981有13条指令，指令是由一个指令代码和一个功能参数组成。指令代码好似参数寄存器的地址代码，而参数才是实质的功能值。MPU向LC7981指令寄存器写入指令代码来选择参数寄存器，再通过数据寄存器向参数寄存器写入参数值，以实现功能的设置。

LC7981向MPU提供一个忙（BF）标志位：

BF = 1表示当前LC7981处于内部运行状态，不接受MPU的访问（读忙标志位除外）；

BF = 0表示LC7981允许MPU的访问。

MPU在访问LC7981时都要判断BF是否为0。

MPU可在RS = 1下从数据总线D7位上读出BF标志值。

下面逐条介绍这些指令的设置：

### 1、方式控制，指令代码为00H

该指令参数定义了显示方式。其参数格式为：

0	0	D5	D4	D3	D2	D1	D0
---	---	----	----	----	----	----	----

D0：字符发生器选择，D0 = 0为CGROM，D0 = 1为EXCGROM；

D1：显示方式选择，D1 = 0为文本方式，D1 = 1为图形方式；

D3与D2组合实现如下功能：

D3	D2	功 能
0	0	光标禁止
0	1	启用光标
1	0	光标禁止，字符闪烁
1	1	光标闪烁

D4：工作方式选择，D4 = 0为从方式，D4 = 1为主方式；

D5：显示状态选择，D5 = 0为禁止显示，D5 = 1为启用显示。

需要注意的是，使用图形方式（D1 = 1）时，只能使D0 = D2 = D3 = 0。一般情况下，LC7981采用主方式（D4 = 1）。当有两片LC7981并联时，则其中一片为主方式，一片为从方式。



## 2、字体设置，指令代码01H

该指令设置文本方式下字符的点阵大小，指令参数格式：

VP - 1	0	HP - 1
--------	---	--------

VP：字符点阵行数，取值范围1~16；

HP：字符点阵列数，图形方式表示一字节显示数据的有效位数，HP的取值范围为6，7，8。

## 3、显示域设置，指令代码02H

该指令的指令参数格式如下：

0	HN - 1
---	--------

HN为一行显示所占的字节数，其取值范围为2~128内的偶数值，由HN和HP可得显示屏有效显示点列数 $N = HN \times HP$ 。

## 4、帧设置，指令代码03H

该指令的指令参数格式如下：

0	NX - 1
---	--------

NX为显示时的帧扫描行数，其倒数即为占空比。对单屏结构的显示模块，NX即为其有效显示行数；对双屏显示结构的模块则2NX为其有效显示行数。

## 5、光标位置设置，指令代码04H

文本方式下的光标为一行（8×1）点阵显示，该指令用来指明光标在字符位中的第几行，指令参数格式如下：

0	0	0	0	CP - 1
---	---	---	---	--------

其中CP表示光标在字符体中的行位置，CP取值范围在1~VP之间。CP>VP时光标将被禁止。

## 6、SADL设置，指令代码08H

该指令的指令参数格式如下：

起始地址低8位	SADL
---------	------

## 7、SADH设置，指令代码09H

该指令的指令参数格式如下：

起始地址高8位	SADH
---------	------

以上两条指令设置了显示缓冲区起始地址，它们的指令参数分别是该地址的低位和高位字节。该地址对应显示屏上左上角显示位，显示缓冲区单元（即RAM单元）与显示屏上的显示位的一一对应关系如下表所示：

SAD	SAD + 1	... ..	SAD + HN - 1
SAD + HN	SAD + HN + 1	... ..	SAD + 2HN - 1
...	...	...	...
...	...	...	...
...	...	...	...





SAD + MHN	SAD + MHN + 1	... ..	SAD + ( M + 1 ) HN - 1
--------------	------------------	--------	------------------------

显示屏上一个显示位长为HP点阵。

## 8、CACL设置，指令代码0AH

该指令的指令参数格式如下：

光标地址低8位 CACL
--------------

## 9、CACH设置，指令代码0BH

该指令的指令参数格式如下：

光标地址高8位 CACH
--------------

以上两条指令设置光标地址指针，它们的指令参数即是该光标地址指针的低位和高位字节。其作用一是用来指示当前要读、写显示缓冲区单元的地址；二是用在文本方式下，指出光标或闪烁字符在显示屏上的位置。

由于光标地址计数器是16位加1计数器，当第N位从1变到0时，会引起第N + 1位自动加1，因此当设置低8位地址时，若最高位MSB从1变为0就会引起高8位地址的最低位LSB加1。因此设置时应先设置低8位，再设置高8位，而且即使只需修改地址低8位，其高位也要跟着重新设置一次，以确保地址指针设置的万无一失。

## 10、数据写，指令代码0CH

该指令将随后写入数据寄存器的数据送入光标地址指针指向的显示RAM单元。光标地址指针将随着每次数据的写入而自动加一修改。该指令功能的终止将由下一条指令的输入来完成。

## 11、数据读，指令代码0DH

该指令代码写入后，紧跟着一次“空读”操作后，则可以连续读出光标地址指针所指向单元的内容。光标地址指针将随着每次数据的读出而自动加一修改，该指令功能的终止将由下一条指令的输入来完成。

## 12、位清零，指令代码0EH

## 13、为置1，指令代码0FH

以上两条指令的功能是将光标地址指针所指向的显示RAM单元中的某一位清零或置1。指令执行一次，光标地址指针自动加一。指令参数格式为：

0	0	0	0	0	NB - 1
---	---	---	---	---	--------

其中NB为要清零或置1的位址，取值1 ~ 8，对应该字节的LSB ~ MSB位。

以上列出了LC7981的所有指令。由于LC7981的指令代码好似参数寄存器的地址代码，所以在写入一个指令代码后向数据口写入的多个数据都将修改该指令代码所指的参数寄存器内的内



容。


## 第四章 内藏LC7981型液晶模块的应用

在内藏控制器型的液晶模块上，已经完成了控制器与液晶驱动器和显示缓冲区的接口工作，留给用户的仅仅是与MPU的接口。因此只需了解LC7981的指令系统及与MPU接口的工作时序，而无需对液晶显示驱动器及其与LC7981的接口作太多了解就可使用内藏LC7981的液晶显示模块。

本节以清达光电提供的演示板为实用电路，说明内藏LC7981液晶显示模块的接口电路。

### 一、直接访问方式接口电路及驱动程序

清达光电提供的演示板接口电路如下图所示：



驱动子程序如下：

```

A15 = /CS ; A9 = R/W ; A8 = RS
DW_ADD EQU 8000H           ; 写数据口地址
DR_ADD EQU 8200H           ; 读数据口地址
CW_ADD EQU 8100H           ; 写指令口地址
CR_ADD EQU 8300H           ; 读状态口地址
COM EQU 30H                ; 指令代码寄存器
DAT EQU 31H                ; 数据寄存器
  
```

#### 1、读状态位子程序

```

PR0: MOV DPTR,#CR_ADD      ; 设置读状态口地址
  
```



```

PR01:  MOVX    A,@DPTR    ; 读状态
        JB     ACC.7,PR01  ; 判状态位为"0"否?
        RET

```

## 2、写指令代码子程序

```

PR1:   LCALL   PR0        ; 写指令入口
        MOV    DPTR,#CW_ADD ; 设置写指令口地址
        MOV    A,COM      ; 取指令代码
        MOVX   @DPTR,A    ; 写入指令代码
        RET

```

## 3、写指令参数和数据子程序

```

PR2:   LCALL   PR0        ; 写数据入口
        MOV    DPTR,#DW_ADD ; 设置写数据口地址

        MOV    A,DAT      ; 取指令参数或显示数据
        MOVX   @DPTR,A    ; 写入参数或数据
        RET

```

## 4、读显示数据子程序


```

PR3:   LCALL   PR0        ; 读数据入口
        MOV    DPTR,#DR_ADD ; 设置读数据口地址
        MOVX   A,@DPTR    ; 读显示数据
        MOV    DAT,A      ; 存数据
        RET

```

## 二、间接控制方式接口电路及驱动程序

间接控制方式就是使用并行接口与内藏LC7981液晶显示模块接口。MPU通过控制并行接口的输出状态实现对LC7981的控制。接口电路如下图所示。由于并行接口是唯一对液晶显示模块接口，所以液晶显示模块的片选直接接地即可。



驱动子程序如下：

```

RS      EQU    P3.2      ; 通道选择信号
RW      EQU    P3.3      ; 读/写选择信号
E       EQU    P3.4      ; 使能信号

```

### 1、读状态位子程序

```

PR0:    SETB    RS        ; RS=1
        SETB    RW        ; R/W=1
PR01:   MOV     P1,#0FFH   ; 置P1口为"1"

        SETB    E         ; E=1
        MOV     A,P1       ; 读状态
        CLR     E         ; E=0
        JB      ACC.7,PR01 ; 判状态位为"0"否?
        RET

```

### 2、写指令代码子程序

```

PR1:    LCALL   PR0        ; 写指令入口
        CLR     RW        ; R/W=0
        MOV     A,COM      ; 取指令代码
        MOV     P1,A       ; 写入P1口
        SETB    E         ; E=1
        CLR     E         ; E=0
        RET

```

### 3、写指令参数和数据子程序



```

PR2:  LCALL  PR0          ; 写数据入口
      CLR   RS            ; RS=0
      CLR   RW            ; R/W=0
      MOV   A,DAT         ; 取指令参数或显示数据
      MOV   P1,A          ; 写入数据
      SETB  E             ; E=1
      CLR   E             ; E=0
      RET

```

#### 4、读数据子程序

```

PR3:  LCALL  PR0          ; 写数据入口
      CLR   RS            ; RS=0
      MOV   P1,#0FFH     ; 置P1口为"1"
      SETB  E             ; E=1
      MOV   A,P1          ; 读数据
      CLR   E             ; E=0
      MOV   DAT,A        ; 存数据
      RET

```

### 三、应用程序

#### 示例一 初始化。

由于LC7981不能“图文并茂”，所以初始化要根据使用需要而设置。  
初始化子程序(文本方式)

```

INT:  MOV   COM,#00H      ; 方式设置 *
      LCALL PR1
      MOV   DAT,#3CH      ; 开显示,主方式,文本方式
      LCALL PR2           ; 开光标闪烁
      MOV   COM,#01H      ; 字符体设置 *
      LCALL PR1
      MOV   DAT,#77H      ; 字符体为8×8点阵
      LCALL PR2
      MOV   COM,#02H      ; 显示域宽度设置 *
      LCALL PR1
      MOV   DAT,#27H      ; 一行占显示RAM 40个字节
      LCALL PR2
      MOV   COM,#03H      ; 帧信号设置 *
      LCALL PR1
      MOV   DAT,#3FH      ; 64行扫描行
      LCALL PR2
      MOV   COM,#04H      ; 光标形状设置
      LCALL PR1
      MOV   DAT,#07H      ; 光标为底线形式

```



```

LCALL PR2
MOV COM,#08H ; 显示起始地址设置
LCALL PR1
MOV DAT,#00H ; 低字节为 00H
LCALL PR2
MOV COM,#09H ; 显示起始地址设置
LCALL PR1
MOV DAT,#00H ; 高字节为 00H
LCALL PR2
RET

```

### 初始化子程序(图形方式)

```

INT:  MOV COM,#00H ; 方式设置
      LCALL PR1
      MOV DAT,#32H ; 开显示,主方式,图形方式
      LCALL PR2
      MOV COM,#01H ; 字符体设置
      LCALL PR1
      MOV DAT,#77H ; 字符体为8×8点阵
      LCALL PR2
      MOV COM,#02H ; 显示域宽度设置 *
      LCALL PR1
      MOV DAT,#27H ; 一行占显示RAM 40个字节
      LCALL PR2
      MOV COM,#03H ; 帧信号设置 *
      LCALL PR1
      MOV DAT,#3FH ; 64行扫描行
      LCALL PR2
      MOV COM,#08H ; 显示起始地址设置
      LCALL PR1
      MOV DAT,#00H ; 低字节为 00H
      LCALL PR2
      MOV COM,#09H ; 显示起始地址设置
      LCALL PR1
      MOV DAT,#00H ; 高字节为 00H
      LCALL PR2
      RET

```

注：\* 标志的指令参数要根据所控制的液晶显示驱动系统的规定而设置

### 初始化示例程序

```

MAIN:  MOV SP,#60H
      ANL P1,#0E3H ; RS=R/W=E=0
      LCALL INT
      LCALL CLEAR

```

**示例二 清显示缓冲区子程序**

```

CLEAR:  MOV     R2,#00H
        MOV     R3,#10H
        MOV     COM,#0AH
        LCALL  PR1
        MOV     DAT,#00H
        LCALL  PR2
        MOV     COM,#0BH
        LCALL  PR1
        MOV     DAT,#00H
        LCALL  PR2
        MOV     COM,#0CH
        LCALL  PR1
        MOV     DAT,#00H
CLR1:   LCALL  PR2
        DJNZ   R2,CLR1
        DJNZ   R3,CLR1
        RET

```

**示例三 西文字符写入**

该示例提供了文本方式下的西文显示的操作，包括显示RAM地址的设置，西文字符显示，光标操作以及读数据操作等。

**(1) 设置文本显示RAM 地址子程序**

```

CUL     EQU     32H           ; 列坐标地址(以字节为单位)
ROW     EQU     34H           ; 行坐标地址(以字符行为单位)
WR_ADD: MOV     A,ROW         ; 取行坐标地址
        MOV     B,#28H        ; 设置显示域宽度(依实际而定)
        MUL     AB            ; 计算光标指针地址
        ADD     A,CUL         ; 加列坐标地址
        MOV     CUL,A         ; 光标低8位暂存CUL
        MOV     A,B
        ADDC    A,#00H
        MOV     ROW,A         ; 光标高8位暂存ROW
        MOV     COM,#0AH      ; 设置光标指针
        LCALL  PR1
        MOV     DAT,CUL
        LCALL  PR2
        MOV     COM,#0BH
        LCALL  PR1
        MOV     DAT,ROW
        LCALL  PR2
        RET

```

**(2) 西文字符组写入子程序**

```

COUNT EQU     35H           ; 写入字符个数

```



```

DPTR    为字符组首地址
WR_C:   PUSH    DPL                ; 暂存DPTR
        PUSH    DPH
        MOV     COM,#0CH           ; 设置写数据指令代码
        LCALL   PR1
        POP     DPH                ; 取出DPTR
        POP     DPL
WR_C1:   CLR     A
        MOVC    A,@A+DPTR         ; 取字符代码
        INC     DPTR              ; 修正DPTR
        PUSH    DPL                ; 暂存DPTR
        PUSH    DPH

        MOV     DAT,A             ; 写入显示数据
        LCALL   PR2
        POP     DPH                ; 取出DPTR
        POP     DPL
        DJNZ    COUNT,WR_C1       ; 循环
        RET

```

#### 西文字符写入演示程序段

```

MOV     CUL,#00H                ; 设置列坐标为00
MOV     ROW,#00H                ; 设置行坐标为00
LCALL   WR_ADD                  ; 调设置文本显示RAM地址子程序
MOV     DPTR,#CTAB              ; 设置字符组首地址
MOV     COUNT,#1AH              ; 写入字符个数为26
LCALL   WR_C                    ; 调写入西文字符组子程序
SJMP    $

```

#### 西文字符表

```

CTAB:DB 53H,75H,70H,70H,6FH,72H,74H,20H,54H,45H,4CH,20H
      DB 3AH,20H,30H,31H,30H,2DH,36H,32H,37H,38H,30H,33H,37H,39H

```

#### 关光标,开闪烁程序(观察显示效果)

```

MOV     COM,#00H
LCALL   PR1
MOV     DAT,#38H
LCALL   PR2
SJMP    $

```

#### 读数据示例程序段

```

MOV     CUL,#00H                ; 设置列坐标地址为00
MOV     ROW,#00H                ; 设置行坐标地址为00
LCALL   WR_ADD                  ; 调文本显示RAM地址子程序
MOV     COM,#0DH                ; 设置读数据指令代码

```





```

        LCALL    PR1
        LCALL    PR3                ; "空读"
LOOPB:  LCALL    PR3                ; 读显示数据
        MOV     A,#3AH              ; 判所读数据是"."否?
        CJNE    A,DAT,LOOPB         ; 不是则再读下一个单元
        MOV     CUL,#0CH            ; 是则将"."改写成 "-"
        MOV     ROW,#00H
        LCALL    WR_ADD

```

#### 单个西文字符写入程序段

```

        MOV     COM,#0CH            ; 设置写数据指令代码
        LCALL    PR1
        MOV     DAT,#2DH            ; 写字符 "-" 代码
        LCALL    PR2
        SJMP     $

```

#### 示例四 汉字字符写入

由于LC7981显示数据传输是从D0位开始的，所以在制作图形时要注意数据生成的顺序。汉字字符的写入本身就是规则图形的写入，而汉字位生成顺序是由D7位开始的，所以程序要有汉字字模结构的转换子程序。该程序使用的汉字库是由清达光电提供的汉字提取软件LATTICE.EXE生成的。

##### 写汉字字符子程序

```

        CUL     EQU     32H          ; 列坐标地址(以字节为单位)
        ROW     EQU     33H          ; 行坐标地址(以点行为单位)
        CODE    EQU     36H          ; 汉字代码(单字节形式)
        COUNT   EQU     35H          ; 计数器
WR_CC:  MOV     A,CODE
        MOV     DPTR,#CCTAB
        MOV     B,#20H
        MUL     AB
        ADD     A,DPL
        PUSH    ACC
        MOV     A,B
        ADDCA,DPH
        PUSH    ACC
        MOV     A,ROW                ; 取行坐标地址
        MOV     B,#28H                ; 设置显示域宽度(依实际而定)
        MUL     AB                    ; 计算光标指针地址
        ADD     A,CUL                ; 加列坐标地址
        MOV     CUL,A                ; 地址低8位暂存CUL
        MOV     A,B
        ADDC    A,#00H
        MOV     ROW,A                ; 地址高8位暂存ROW
        MOV     COUNT,#00H           ; 计数器清零

```



```

WR_CC1:  MOV     COM,#0AH           ; 设置光标地址
          LCALL   PR1
          MOV     DAT,CUL
          LCALL   PR2
          MOV     COM,#0BH
          LCALL   PR1
          MOV     DAT,ROW
          LCALL   PR2
          MOV     COM,#0CH           ; 写数据指令代码
          LCALL   PR1
          POP     DPH
          POP     DPL               ; 设置汉字库首址
          MOV     A,COUNT
          MOVC    A,@A+DPTR         ; 取汉字左半部字模数据
          MOV     DAT,A             ; 写入字模数据
          PUSH    DPL
          PUSH    DPH
          LCALL   CONVERT           ; 调字模结构转换子程序
          LCALL   PR2
          POP     DPH
          POP     DPL
          MOV     A,COUNT
          ADD     A,#10H             ; 修改间址参数
          MOVC    A,@A+DPTR         ; 取汉字右半部字模数据
          MOV     DAT,A             ; 写入字模数据
          PUSH    DPL
          PUSH    DPH
          LCALL   CONVERT           ; 调字模结构转换子程序
          LCALL   PR2
          MOV     A,CUL             ; 修正光标指针地址
          ADD     A,#28H             ; 加显示域宽度(依实际而定)
          MOV     CUL,A
          MOV     A,ROW
          ADDC    A,#00H
          MOV     ROW,A
          INC     COUNT             ; 计数器加一
          MOV     A,COUNT
          CJNE    A,#10H,WR_CC1
          POP     ACC
          POP     ACC
          RET

```

#### 汉字字模结构转换子程序

```

CONVERT: MOV     A,COUNT           ; 暂存COUNT计数器
          PUSH    ACC
          MOV     COUNT,#08H       ; 借用COUNT作为计数器

```



```

MOV      A,DAT          ; 取数据
MOV      DAT,#00H       ; DAT清零作为新数据
CONVER1:RRC      A      ; ACC.0  C
XCH      A,DAT          ; 新旧数据交换
RLC      A              ; C  ACC.0
XCH      A,DAT          ; 新旧数据交换
DJNZ     COUNT,CONVER1  ; 循环
POP      ACC            ; 恢复COUNT
MOV      COUNT,A
RET

```

### 汉字库

CCTAB: DB 010H,022H,064H,00CH,0C0H,044H,054H,054H;清00

DB 054H,07FH,054H,054H,0D6H,044H,040H,000H

DB 004H,004H,0FEH,001H,000H,000H,0FFH,015H

DB 015H,015H,055H,095H,07FH,001H,000H,000H

DB 040H,042H,0CCH,000H,020H,020H,020H,0A0H;达01

DB 07FH,020H,020H,020H,030H,020H,000H,000H

DB 040H,020H,01FH,020H,050H,048H,046H,041H

DB 040H,041H,042H,04CH,058H,060H,020H,000H

DB 040H,040H,042H,044H,058H,0C0H,040H,07FH;光02

DB 040H,0C0H,050H,048H,046H,064H,040H,000H

DB 000H,080H,040H,020H,018H,007H,000H,000H

DB 000H,03FH,040H,040H,040H,040H,070H,000H

DB 000H,0F8H,048H,048H,048H,048H,0FFH,048H;电03

DB 048H,048H,048H,0FCH,008H,000H,000H,000H

DB 000H,007H,002H,002H,002H,002H,03FH,042H

DB 042H,042H,042H,047H,040H,070H,000H,000H

### 汉字显示演示程序

```

MOV      CUL,#00H
MOV      ROW,#00H
MOV      CODE,#00H
LCALL    WR_CC
MOV      CUL,#02H
MOV      ROW,#08H
MOV      CODE,#01H
LCALL    WR_CC
MOV      CUL,#04H
MOV      ROW,#10H
MOV      CODE,#02H
LCALL    WR_CC

```



```

MOV      CUL,#06H
MOV      ROW,#18H
MOV      CODE,#03H
LCALL    WR_CC
SJMP     $

```

## 第五章 液晶模块的调试经验

下面是清达光电的开发和技术服务人员在内部调试和协助用户调试过程中积累的一些经验，希望能对您做调试工作有所帮助，同时希望能与您共享您的调试经验。

- 1、 MPU到液晶模块接口的连线尽可能短，最好5厘米之内，如果系统要求必须超过5厘米，就要考虑加驱动，例如MPU信号线上接74LS245。
- 2、 在VDD、V0/VEE对地（VSS）间接0.1u（104）左右电容去耦，接10u或20u电容滤波。
- 3、 对于一些不好找干扰源的系统，可以考虑用软件“掩饰”显示不正常的问题，即定时初始化控制器，刷新数据。
- 4、 SED1520的上电复位过程将迟于MPU的复位时间，在程序上电执行液晶模块初始化程序之前先运行一段延时程序。
- 5、 将A0、R/W、CS、E这些功能信号线置于地址线高位（例如8031的P2口）。
- 6、 间接控制方式驱动程序的编制要求E信号独立指令操作，不要与其它信号合成操作。
- 7、 时序改造芯片选择TTL电平的，不要选择CMOS电平的，例如我们上面举例的电路就要选74LS00，而不要选74HC00，更不能选4000系列。
- 8、 如果选用的MPU为高速器件，或者它的总线读写周期小于1us，建议采用间接方式，即模拟液晶控制器时序的方式与液晶控制器通信。
- 9、 如果用在临界时序时，出现有时显示正常有时显示不正常的现象，或者上批显示正常，这批显示不正常，可以首先检查转换电路的结构，还可考虑在E、R/W、A0信号处分别对地接0.001u(102)的电容试试，即调整这三个信号的相对时间间隔看看。
- 10、 如果模块（例如122\*32液晶）68/80时序可选，最好选与MPU匹配的时序。
- 11、 有问题，请及时与我们的技术服务人员联系。

### 附录一 清达光电配套产品

1. 各种液晶模块的底层演示板。
2. 负电压产生模块DC-DC变换器。
3. 各种液晶显示器件背光所需的逆变器，有配EL背光和配CCFL背光的。

### 附录二 清达光电维修服务规范

随着LCM的应用越来越广泛，LCM的售后维修服务工作越来越重要。为了能长期为用户提供良好的售后服务，公司特制定有关LCM的维修范围及维修项目的收费标准。

1. LCM不属于设备，而是作为元器件提供给用户。因此购买LCM的客户不享受保修服务。



2. 无偿维修服务：由于产品质量造成LCM显示不正常，及外观缺陷，公司将提供免费维修，必要时可调换商品。
3. 有偿维修服务：凡由于人为造成LCM的外观及电路的损坏，以至造成LCM无法正常显示，客户应承担一定数额的维修成本费用。
4. 本公司自行设计、制做的各种板、卡的维修服务，具体办法参照LCM的规范。
5. 对于本公司销售的LCM及相关产品，客户在使用过程中一旦发现问题，请及时与本公司联系，切勿自行处理。

### 附录三 LCM的常见故障

项目	故障现象
液晶片	缺行/列；共用极/图画极短路；字暗/行暗；液晶外观问题；电压不当；多字画；显示模糊；彩虹；气泡；黑点；针孔；液晶片损坏；偏振片擦花；玻璃碎/玻璃裂
线路板	线路板弯曲；线路板短路、断路；电镀孔断路；线路损坏；焊盘损坏
背光	背光不均匀；扩散膜不平；背光板损坏；背光灯损坏；背光逆变器损坏
斑马条	斑马条扭曲；斑马条移位
热压导电膜	热压片损坏；热压片焊接不良
外框	外框损坏；外框松动
芯片	IC失效；IC管脚中有异物